// Upgrade NOTE: replaced '_Object2World' with 'unity_ObjectToWorld'

Shader "UShaderBook/C6/SpecularVertexLevel"
{
Properties
    {
        _Diffuse ("Diffuse", Color) = (1,1,1,1)
        _Specular ("Specular", Color) = (1,1,1,1)
        _Gross ("Gross", Range(8.0, 256.0)) = 20.0
    }
    SubShader
    {
        Pass{
            Tags {"LightMode"="ForwardBase"}
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #include "Lighting.cginc"

            fixed4 _Diffuse;
            fixed4 _Specular;
            float _Gross;

            struct a2v{
                float4 vertex : POSITION;
                float3 normal : NORMAL; 
            };

            struct v2f{
                float4 pos : SV_POSITION;
                fixed3 color: COLOR;// 顶点着色器不再需要计算颜色，而是把世界空间法线传给片元着色器
            };

            v2f vert(a2v v){
                v2f o;
                o.pos = UnityObjectToClipPos(v.vertex);
                fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz; // 环境光颜色获取
                // 漫反射 diffuse
                fixed3 worldNormal = normalize(UnityObjectToWorldNormal(v.normal));
                fixed3 worldLightDir = normalize(_WorldSpaceLightPos0.xyz); // 光源方向

                fixed3 diffuse = _LightColor0.rgb * _Diffuse.rgb * saturate(dot(worldNormal, worldLightDir));
                // 漫反射
                // 高光反射 specular
                fixed3 reflectDir = normalize(reflect(-worldLightDir, worldNormal)); // 反射方向
                fixed3 viewDir = normalize(_WorldSpaceCameraPos.xyz - mul(unity_ObjectToWorld, v.vertex)); // 顶点到相机的方向
                fixed3 specular = _LightColor0.rgb * _Specular.rgb * pow(saturate(dot(reflectDir, viewDir)), _Gross);
                
                o.color = ambient + diffuse + specular;
                return o;
            }

            float4 frag(v2f i) : SV_TARGET{
                return fixed4(i.color, 1.0);
            }
            ENDCG
        }
    }
    Fallback "Specular"
}
